



Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

**A MODULAR THREE-DIMENSIONAL
FINITE-DIFFERENCE GROUND-WATER
FLOW MODEL**

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological
Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

storage term maintains the flexibility of the program to represent layers of varying thickness, and to implement either the direct three-dimensional or "quasi-three-dimensional" conceptualizations of vertical discretization. The storage coefficient values are read layer by layer; they are designated as array `sfl` in the input instructions. These values are then multiplied by the cell areas, $\Delta r_j \Delta c_j$, to create storage capacity values, and they are stored in the `SC1` array.

Storage Term Conversion

The primary storage capacity described above, $SC1_{i,j,k}$ is adequate for simulations in which the water level in each individual cell remains either above the top of the cell or below the top of the cell throughout the course of the simulation. If the water level crosses the top of a cell during a simulation--i.e., if the water level in a confined (fully saturated) cell falls below the top of the cell as a result of simulated pumpage, or if the water level in an unconfined cell rises above the top of the cell--then in effect the system "converts" from confined to water table conditions, or vice versa, during the simulation. Where these conditions appear to be possible, the user may invoke storage term conversion for the entire layer through use of the layer-type flag. When this is done, the primary storage capacity, $SC1_{i,j,k}$ for any cell in the layer will represent the confined storage coefficient multiplied by cell area; a secondary storage capacity, $SC2_{i,j,k}$ is used to represent specific yield multiplied by cell area. Values of confined storage coefficient for each cell in the layer are read through the two-dimensional input array `sfl`. These confined storage

coefficient values are multiplied by cell areas to obtain confined storage capacities, which are stored in the array SC1. Values of specific yield for each cell in the layer are read through the two-dimensional input array sf2. These specific yield values are multiplied by cell areas to obtain unconfined storage capacities, which are stored in array SC2.

In a layer which has been designated for storage term conversion, the expression for rate of accumulation in storage in cell i,j,k is formulated as follows

$$\frac{\Delta V}{\Delta t} = \frac{SCB (h_{i,j,k}^m - TOP_{i,j,k}) + SCA (TOP_{i,j,k} - h_{i,j,k}^{m-1})}{t_m - t_{m-1}} \quad (61)$$

where again $\frac{\Delta V}{\Delta t}$ is rate of accumulation of water in storage in cell i,j,k and as such must appear on the right side of equation (24) or (54); SCA is the storage capacity in effect in cell i,j,k at the start of the time step; and SCB is the "current" storage capacity--that is, the storage capacity in effect during the iteration in process. Consider a case in which the head in cell i,j,k at the beginning of time step m ($h_{i,j,k}^{m-1}$) is above the top of the cell. Since there is no free surface in the cell at the start of the time step, the storage capacity at that time is taken as the confined storage capacity--that is, SCA is set equal to $SC1_{i,j,k}$. If, during a given iteration for time step m, the computed value of head for the end of the time step ($h_{i,j,k}^m$) is found to be above the top of the cell, SCB for the following iteration is also set equal to $SC1_{i,j,k}$; equation (61) for that

iteration then reverts to the form of equation (60). However, if the computed value of $h_{i,j,k}^m$ in a given iteration turns out to be below the top of the cell, as shown in figure 30, the value of SCB for the following iteration is set equal to SC2, the unconfined storage capacity. In this case the computed rate of release of water from storage in the time step has two components:

$$SC1_{i,j,k} (TOP_{i,j,k} - h_{i,j,k}^{m-1}) / (t_m - t_{m-1}) ,$$

the rate of release from confined or compressive storage; and

$$SC2_{i,j,k} (h_{i,j,k}^m - TOP_{i,j,k}) / (t_m - t_{m-1}) ,$$

the rate of release from water table storage.

If the head at the beginning of the time step, $h_{i,j,k}^{m-1}$, is below the top of cell i,j,k , so that a free surface exists within the cell, SCA in equation (61) is set equal to $SC2_{i,j,k}$. If, during an iteration for time step m , the computed value of head for the end of the time step turns out to be below the top of the cell, SCB in the subsequent iteration is also set equal to $SC2_{i,j,k}$ and equation (61) again reverts to the form of equation (60). However, if the computed head for the end of the time step turns out to be above the top of the cell, SCB in the subsequent iteration is set equal to $SC1_{i,j,k}$, the confined storage capacity. This situation occurs during intervals of rising water level, and again two components are computed for the rate of accumulation of water in storage--one corresponding to unconfined or water table storage and one corresponding to confined or compressive storage.

Equation (61) can be rearranged as follows

$$\frac{\Delta V}{\Delta t} = \frac{SCB}{t_m - t_{m-1}} h_{i,j,k}^m + \frac{SCA (TOP_{i,j,k} - h_{i,j,k}^{m-1}) - SCB * TOP_{i,j,k}}{t_m - t_{m-1}} \quad (62)$$




Figure 30.—A model cell which uses two storage factors during one iteration.

Again, $\frac{\Delta V}{\Delta t}$ represents rate of accumulation in storage and as such would appear on the right in equation (24) or (54). In the formulation of equation (26), therefore, the term $\frac{SCB}{t_m - t_{m-1}}$ is subtracted from $HCOF_{i,j,k}$ on the left hand side, while the term $\frac{SCA(TOP_{i,j,k} - h_{i,j,k}^{m-1}) - SCB*TOP_{i,j,k}}{t_m - t_{m-1}}$ is added to $RHS_{i,j,k}$ on the right.

Applicability and Limitations of Optional Formulations

The options for calculation of horizontal conductance under water table conditions, limitation of vertical flow under desaturating conditions, and storage term conversion were all developed on the assumption that each model layer corresponds to a distinct aquifer or permeable horizon, and that these horizons are separated by distinct units of low permeability. Use of these options where these conditions are not satisfied may lead to a variety of problems and inaccuracies in simulation. For example, if the option for horizontal conductance calculation under water table conditions is used where a water table aquifer is represented by several model layers, and the water table is expected to traverse more than one layer during simulation, incorrect (and irreversible) conversion of cells to a no-flow condition may occur as iterations are carried out. Thus care should be exercised in the decision to use any of the three options noted above.

Data Requirements

The fundamental variables controlling cell-to-cell flow and storage in the model are entered through the Block-Centered-Flow Package input. These variables, depending on the options which are invoked, may include

transmissivity, hydraulic conductivity, specific yield, confined storage coefficient, vertical leakance, aquifer bottom elevation and aquifer top elevation. Each of these variables is entered using the utility array-reader module U2DREL, which is described in Chapter 14. This module either reads a two-dimensional array of data for a single layer, or accepts a single value provided by the user and applies that value throughout the array, for all cells in the layer.

The model utilizes a layer-type code to classify layers according to the simulation options that are used. In particular, the layer-type code indicates whether specified transmissivity values are to be used, or transmissivities are to be calculated at each iteration as the product of hydraulic conductivity and saturated thickness; whether storage term conversion is to be used; and whether limitation of vertical flow from above is to be invoked under dewatered conditions. Because the layer-type code identifies the options to be employed in a given layer, it indicates the kinds of data required for the layer, and thus identifies the data arrays to be read. The data are entered layer by layer; for each layer a set of two-dimensional arrays, one array for each required parameter, is read in turn. That is, all of the required arrays for layer 1 are read initially, in sequence, then all of the arrays for layer 2, and so on until all layers have been covered. This method of data organization provides a simpler input process than would be possible using the alternative of a series of three-dimensional arrays corresponding to the various parameters.

Within each layer the required parameters should be specified for every cell, including constant-head and no-flow cells. For no-flow cells,

the entered values are never used in calculation, and thus any values may be specified; for constant head cells, the storage terms are not used but the other parameters are, and realistic values for those parameters must be entered.

Two parameters, transmissivity and hydraulic conductivity, each require the designation of two values at each cell--one in the row direction and one in the column direction. To reduce input effort, only a single array is read for each of these parameters, giving only the values in the row direction; these row-direction values are subsequently multiplied by an anisotropy factor to obtain the corresponding column-direction values. A single value of the anisotropy factor is specified by the user for each layer, through the one-dimensional array TRPY (NLAY).

Vertical leakance terms (V_{cont} , or $K_z/\Delta z$) are associated with each layer except the lowermost; the values associated with a given layer actually apply to the interval between that layer and the next lower layer. For example, the array of V_{cont} values entered during the input sequence for layer 1 actually applies to the interval between the midpoint of layer 1 and the midpoint of layer 2.

In addition to the terms mentioned above, the Block-Centered Flow Package input includes cell dimensions (DELR and DELC), a flag to indicate whether the simulation is transient or steady state (ISS), and a flag to indicate whether cell-by-cell flow terms are to be saved (IBCFCB). If the ISS flag is set for steady-state conditions ($ISS \neq 0$), no space is allocated for storage coefficient or specific yield and storage calculations are skipped. Thus for steady-state runs, arrays of storage coefficients or

specific yields must not be included in the input data; if they are included, the data sequence will be misread. Note that erroneous specification of ISS or of a LAYCON value will also cause misreading of the data array sequence.

Four types of layer are recognized by the model, incorporating various combinations of the options provided by the Block-Centered-Flow Package. These four layer types are identified by their layer-type codes, which are stored in the one-dimensional array LAYCON (NLAY). The code values and the corresponding layer characteristics are given below.

Layer-type 0--In this category there is no provision for modification of transmissivity as water level varies, for storage term conversion, or for limitation of vertical flow from above if water level falls below the top of the cell. This layer type is normally used to simulate confined conditions, but could also be used to simulate a layer in which unconfined conditions will always prevail, provided drawdowns are expected to be a small fraction of layer thickness and flow from the overlying layer (if present) is expected to be negligible. If the simulation is transient, storage coefficient or specific yield values are entered in the input array `sf1(NCOL, NROW)`; then row-direction transmissivities are entered in the input array `Tran (NCOL, NROW)`; and following the transmissivities, unless the layer is the lowermost in the model, vertical leakance values are entered in the input array `Vcont (NCOL, NROW)`. Again, parameter values may be specified by providing the entire array, or by providing a single default value which is applied to all cells of the layer. The parameter values assigned at the beginning of a simulation in this type of layer are retained without change throughout the simulation.

Layer-type 1--This layer type is utilized only in a single-layer model or in the uppermost layer of a model, and only where unconfined conditions are expected to persist in the layer throughout the entire period of simulation. No provision is made for storage term conversion, by virtue of the assumption that water table conditions will always prevail; and no provision is made for limiting flow from above under dewatered conditions, since layer-type 1 is used only for the uppermost layer of a model. However, transmissivities are computed at each iteration as the products of hydraulic conductivity and saturated thickness values within the layer. Thus the input data includes hydraulic conductivity and cell bottom elevation, rather than transmissivity. If the simulation is indicated as transient, specific yield values are entered in the input array `sf1(NCOL, NROW)`. Row direction hydraulic conductivity values are then entered in the input array `HY(NCOL, NROW)` and cell bottom elevations are entered in the array `BOT(NCOL, NROW)`. If the model contains more than one layer, vertical leakage values are entered in the input array `Vcont(NCOL, NROW)`. Because use of this layer type would be inappropriate except in the uppermost layer, a check of the layer number is made whenever `LAYCON` is given a value of one; if the layer number is not also equal to one, indicating the uppermost model layer, an error message is printed.

Layer-type 2--This layer type is used where the situation may alternate between confined and unconfined conditions, so that storage term conversion and limitation of flow from above under dewatered conditions are both desirable; but where the saturated thickness is expected to remain everywhere a high fraction of the layer thickness throughout the period of simulation,

so that recalculation of transmissivity as the product of hydraulic conductivity and saturated thickness is not necessary. The storage term conversion option requires that both a confined storage coefficient and a specific yield value be specified for each cell, and that the top elevation be specified for each cell; the top elevation is also used in the option to limit flow from above under dewatered conditions. If the simulation is transient, confined storage coefficient values are entered in the input array `sf1(NCOL, NROW)`. Transmissivity values are then entered in the array `Tran(NCOL, NROW)`. Unless the layer is the lowermost in the model, vertical leakance values are next entered in the array `Vcont(NCOL, NROW)`. Specific yield values are then entered in the array `sf2(NCOL, NROW)` if the simulation is transient; and finally layer top elevations are entered in the array `TOP (NCOL, NROW)`.

Layer-Type 3--This layer type incorporates all of the Block-Centered-Flow options associated with water table conditions. Transmissivities are recalculated at each iteration using hydraulic conductivities and layer bottom elevations, and both storage term conversion and limitation of flow from above under dewatered conditions are implemented. The required data thus includes hydraulic conductivities, layer bottom elevations, confined storage coefficients (if transient), specific yields (if transient), vertical leakances and layer top elevations. Confined storage coefficients are entered in the input array `sf1 (NCOL, NROW)`; hydraulic conductivity values are then entered in the array `HY(NCOL, NROW)`, and aquifer bottom elevations in `BOT(NCOL, NROW)`. Unless the layer is the lowermost in the model, vertical leakance values are next entered in the array `Vcont(NCOL, NROW)`. Specific yield values are then entered in the array

sf2(NCOL,NROW); and finally aquifer top elevations are entered in the array TOP(NCOL, NROW).

The input sequence is outlined in the following section. Both of the utility modules which are used are described in Chapter 14, and the required formats are illustrated in the "Sample Input to the BCF Package" and in appendix D.

Block-Centered Flow Package Input

Input for the Block-Centered Flow (BCF) Package is read from the unit specified in IUNIT(1).

FOR EACH SIMULATION

BCF1AL

1. Data: ISS IBCFCB
Format: I10 I10
2. Data: LAYCON(NLAY) (Maximum of 80 layers)
Format: 40I2

(If there are 40 or fewer layers, use one record; otherwise, use two records.)

BCF1RP

3. Data: TRPY(NLAY)
Module: U1DREL
4. Data: DELR(NCOL)
Module: U1DREL
5. Data: DELC(NROW)
Module: U1DREL

A subset of the following two-dimensional arrays are used to describe each layer. The arrays needed for each layer depend on the layer type code (LAYCON) and whether the simulation is transient (ISS = 0) or steady state (ISS ≠ 0). If an array is not needed, it must be omitted. All of the arrays (items 6-12) for layer 1 are read first; then all of the arrays for layer 2, etc.

IF THE SIMULATION IS TRANSIENT

6. Data: sf1(NCOL,NROW)
Module: U2DREL

IF THE LAYER TYPE CODE (LAYCON) IS ZERO OR TWO

7. Data: Tran(NCOL,NROW)
Module: U2DREL

IF THE LAYER TYPE CODE (LAYCON) IS ONE OR THREE

8. Data: HY(NCOL,NROW)
Module: U2DREL
9. Data: BOT(NCOL,NROW)
Module: U2DREL

IF THIS IS NOT THE BOTTOM LAYER

10. Data: Vcont(NCOL,NROW)
Module: U2DREL

IF THE SIMULATION IS TRANSIENT AND THE LAYER TYPE CODE (LAYCON) IS TWO OR THREE

11. Data: sf2(NCOL,NROW)
Module: U2DREL

IF THE LAYER TYPE CODE IS TWO OR THREE

12. Data: TOP(NCOL,NROW)
Module: U2DREL

Explanation of Fields Used in Input Instructions

ISS--is the steady-state flag.

If ISS \neq 0, the simulation is steady state.

If ISS = 0, the simulation is transient.

IBCFCB--is a flag and a unit number.

If IBCFCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set; the terms which are saved will include cell-by-cell storage terms, cell-by-cell constant head flows, and internal cell-by-cell flows.

If IBCFCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IBCFCB < 0, flow for each constant-head cell will be printed, rather than saved on disk, whenever ICBCFL is set; cell-by-cell storage terms and internal cell-by-cell flows will neither be saved nor printed.

LAYCON--is the layer type table. Each element holds the code for the respective layer. Read one value for each layer. There is a limit of 80 layers. Leave unused elements blank.

0 - confined--Transmissivity and storage coefficient of the layer are constant for the entire simulation.

1 - unconfined--Transmissivity of the layer varies. It is calculated from the saturated thickness and hydraulic conductivity. The storage coefficient is constant; valid only for layer 1.

2 - confined/unconfined--Transmissivity of the layer is constant. The storage coefficient may alternate between confined and unconfined values. Vertical leakage from above is limited if the layer desaturates.

3 - confined/unconfined--Transmissivity of the layer varies. It is calculated from the saturated thickness and hydraulic conductivity. The storage coefficient may alternate between confined and unconfined values. Vertical leakage from above is limited if the aquifer desaturates.

TRPY--is a one-dimensional array containing an anisotropy factor for each layer. It is the ratio of transmissivity or hydraulic conductivity (whichever is being used) along a column to transmissivity or hydraulic conductivity along a row. Read one value per layer. Set to 1.0 for isotropic conditions. NOTE: This is one array with one value for each layer.

DELR--is the cell width along rows. Read one value for each of the NCOL columns.

DELC--is the cell width along columns. Read one value for each of the NROW rows.

sf1--is the primary storage coefficient. Read only for a transient simulation (steady-state flag, ISS, is 0). Note that for Laycon=1, sf1 will always be specific yield, while for Laycon=2 or 3, sf1 will always be confined storage coefficient. For Laycon=0, sf1 would normally be confined storage coefficient; however, layer-type 0 can also be used for simulation of water table conditions where drawdowns are expected to remain everywhere a small fraction of the saturated thickness, and where there is no layer above, or flow from the layer above is negligible; and in this case specific yield values would be entered in sf1.

Tran--is the transmissivity along rows. Tran is multiplied by TRPY to obtain transmissivity along columns. Read only for layers where LAYCON is zero or two.

HY--is the hydraulic conductivity along rows. HY is multiplied by TRPY to obtain the hydraulic conductivity along columns. Read only for layers where LAYCON is one or three.

BOT--is the elevation of the aquifer bottom. Read only for layers where LAYCON is one or three.

Vcont--is the vertical hydraulic conductivity divided by the thickness from a layer to the layer beneath it. Since there is not a layer beneath the bottom layer, Vcont cannot be specified for the bottom layer.

sf2--is the secondary storage coefficient. Read it only for layers where LAYCON is two or three and only if a transient simulation (steady-state flag, ISS, is zero). The secondary storage coefficient is always specific yield.

TOP--is the elevation of the aquifer top. Read only for layers where LAYCON is two or three.

Module Documentation for the Block-Centered Flow Package


The Block-Centered Flow Package (BCF1) has four primary modules and three submodules. The relationship of the modules to MAIN and to each other is shown in figure 31. The flow of information used to calculate horizontal-hydraulic conductances (CC and CR) is shown for several of the modules. For example, BCF1RP passes transmissivity (T) and cell dimensions (DELR and DELC) to SBCF1N. Module SBCF1N then returns CC and CR to BCF1RP. The modules are:

Primary Modules

BCF1AL	Allocates space for data arrays.
BCF1RP	Reads all data needed by the package, invokes SBCF1N to reconcile input transmissive values with the IBOUND array, and calculates storage capacities and constant conductances.
BCF1FM	Calculates all coefficients of the system of equations that are not constant and invokes SBCF1H to calculate horizontal-branch conductances in partially saturated layers.
BCF1BD	Calculates flow rates and accumulated flow volumes into and out of storage and constant-head boundaries. When cell-by-cell flow is specified, flow across all sides of each cell is also calculated.

Submodules

SBCF1N	Reconciles input transmissive values with the IBOUND array and calculates storage capacities and constant conductances. Invokes SBCF1C to calculate horizontal-branch conductances for layers where transmissivity is constant.
SBCF1H	Calculates transmissivity for cells in layers where it depends on heads and invokes SBCF1C to calculate horizontal-branch conductances.
SBCF1C	Calculates horizontal-branch conductance from cell transmissivity.
SBCF1B	Calculates cell-by-cell flow terms across cell faces.
SBCF1F	Calculates flow terms (both cell-by-cell and entries to overall budget) for flow to and from constant-head cells.



Explanation

CC	Conductance in the Column Direction	HY	Hydraulic Conductivity
CR	Conductance in the Row Direction	TOP	Elevation of the Top of a Layer
T	Transmissivity	BOT	Elevation of the Bottom of a Layer
DELR	Grid Spacing in the Row Direction	DELC	Grid Spacing in the Column Direction

Figure 31.—Relationship among the modules in the Block-Centered Flow Package.

Narrative for Module BCF1AL

This module allocates space for data arrays for the Block-Centered Flow Package. It is done in the following order:

1. Print the message identifying the package.
2. Read and print the steady-state flag ISS and the cell-by-cell flow-term unit and flag (IBCFB). Cell-by-cell flow terms for the BCF Package are flow to the right, flow forward, flow down, increase in storage, and flow to constant heads.
3. Read and print the layer-type code and count the number of layers which need the TOP array and the BOTTOM array.
 - (a) Read the layer-type codes.
 - 0 = confined
 - 1 = unconfined
 - 2 = confined/unconfined but transmissivity is constant
 - 3 = confined/unconfined but transmissivity depends on head
 - (b) Initialize the counters KT and KB in which the numbers of layers needing the TOP and BOTTOM are accumulated.
 - (c) For each layer, print the layer-type code and determine if TOP and/or BOTTOM arrays are needed.
 - (1) Print the layer number and the layer-type code.
 - (2) If a layer other than the top layer is unconfined (type = 1), print an error message and STOP.

(3) If the layer type is one or three, add one to the BOTTOM counter, KB.

(4) If the layer type is two or three, add one to the TOP counter, KT.

4. Calculate the number of elements in the grid and in a layer.

5. Allocate space for the following arrays:

SC1 Primary storage capacity;
SC2 Secondary-storage capacity (layer type 2 or 3 only);
TRPY Horizontal anisotropy factor;
BOT Bottom of layers (layer type 2 or 3 only);
TOP Top of layers (layer type 2 or 3 only); and
HY Hydraulic conductivity (layer type 1 or 3 only).

The following notes apply:

If the simulation is transient (ISS = 0), storage coefficients are needed.

The number of vertical conductance arrays is one less than the number of layers.

6. Print the amount of space used by the BCF Package.

7. RETURN.

Flow Chart for Module BCF1AL

ISS is the steady-state flag. If it is set (ISS = 1), the simulation is steady state (storage is not considered).

IBCFCB is a flag and a unit number.

If IBCFCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If IBCFCB = 0, cell-by-cell flow terms will not be printed or recorded.


If IBCFCB < 0, flow from constant-head cells will be printed whenever ICBCFL is set.

LAYCON is a layer-type code (one for each layer).

- 0 - confined
- 1 - unconfined
- 2 - confined/unconfined but transmissivity is constant
- 3 - confined/unconfined

KT is a counter for the number of layers for which TOP is needed. (It is also the number of layers for which a secondary storage factor is needed.)

KB is a counter for the number of layers for which BOTTOM is needed. (It is also the number of layers for which hydraulic conductivity is needed.)



```

SUBROUTINE BCF1AL (ISUM, LENX, LCSC1, LCHY, LCBOT,
1      LCTOP, LCSC2, LCTRPY, IN, ISS, NCOL, NROW, NLAY, IOUT, IBCFCB)
C
C-----VERSION 1542 12MAY1987 BCF1AL
C
C *****
C ALLOCATE ARRAY STORAGE FOR BLOCK-CENTERED FLOW PACKAGE
C *****
C
C      SPECIFICATIONS:
C -----
C      COMMON /FLWCOM/LAYCON(80)
C -----
C
C1-----IDENTIFY PACKAGE
      WRITE(IOUT,1)IN
      1 FORMAT(1H0,'BCF1 -- BLOCK-CENTERED FLOW PACKAGE, VERSION 1',
      1', 9/1/87',' INPUT READ FROM UNIT',I3)
C
C2-----READ AND PRINT ISS (STEADY-STATE FLAG) AND IBCFCB (FLAG FOR
C2-----PRINTING OR UNIT# FOR RECORDING CELL-BY-CELL FLOW TERMS)
      READ(IN,2) ISS,IBCFCB
      2 FORMAT(2I10)
      IF(ISS.EQ.0) WRITE(IOUT,3)
      3 FORMAT(1X,'TRANSIENT SIMULATION')
      IF(ISS.NE.0) WRITE(IOUT,4)
      4 FORMAT(1X,'STEADY-STATE SIMULATION')
      IF(IBCFCB.GT.0) WRITE(IOUT,9) IBCFCB
      9 FORMAT(1X,'CELL-BY-CELL FLOWS WILL BE RECORDED ON UNIT',I3)
      IF(IBCFCB.LT.0) WRITE(IOUT,88)
      88 FORMAT(1X,'CONSTANT HEAD CELL-BY-CELL FLOWS WILL BE PRINTED')
C
C3-----READ TYPE CODE FOR EACH LAYER AND COUNT TOPS AND BOTTOMS
      IF(NLAY.LE.80) GO TO 50
      WRITE(IOUT,11)
      11 FORMAT(1H0,'YOU HAVE SPECIFIED MORE THAN 80 MODEL LAYERS'/1X,
      1 'SPACE IS RESERVED FOR A MAXIMUM OF 80 LAYERS IN ARRAY LAYCON')
      STOP
C
C3A-----READ LAYER TYPE CODES.
      50 READ(IN,51) (LAYCON(I),I=1,NLAY)
      51 FORMAT(40I2)
C      BOTTOM IS READ FOR TYPES 1,3      TOP IS READ FOR TYPES 2,3
      WRITE(IOUT,52)
      52 FORMAT(1X,5X,'LAYER AQUIFER TYPE',/1X,5X,19(' -'))
C
C3B-----INITIALIZE TOP AND BOTTOM COUNTERS.
      NBOT=0
      NTOP=0
C
C3C-----PRINT LAYER TYPE AND COUNT TOPS AND BOTTOMS NEEDED.
      DO 100 I=1,NLAY
C

```

```

C3C1-----PRINT LAYER NUMBER AND LAYER TYPE CODE.
      L=LAYCON(I)
      WRITE(IOUT,7) I,L
      7 FORMAT(1X,I9,I10)
C
C3C2-----ONLY THE TOP LAYER CAN BE UNCONFINED(LAYCON=1).
      IF(L.NE.1 .OR. I.EQ.1) GO TO 70
      WRITE(IOUT,8)
      8 FORMAT(1H0,'AQUIFER TYPE 1 IS ONLY ALLOWED IN TOP LAYER')
      STOP
C
C3C3-----LAYER TYPES 1 AND 3 NEED A BOTTOM. ADD 1 TO KB.
      70 IF(L.EQ.1 .OR. L.EQ.3) NBOT=NBOT+1
C
C3C4-----LAYER TYPES 2 AND 3 NEED A TOP. ADD 1 TO KT.
      IF(L.EQ.2 .OR. L.EQ.3) NTOP=NTOP+1
      100 CONTINUE
C
C
C
C4-----COMPUTE DIMENSIONS FOR ARRAYS.
      NRC=NROW*NCOL
      ISIZ=NRC*NLAY
C
C5-----ALLOCATE SPACE FOR ARRAYS. IF RUN IS TRANSIENT(ISS=0)
C5-----THEN SPACE MUST BE ALLOCATED FOR STORAGE.
      ISOLD=ISUM
      LCSC1=ISUM
      IF(ISS.EQ.0) ISUM=ISUM+ISIZ
      LCSC2=ISUM
      IF(ISS.EQ.0) ISUM=ISUM+NRC*NTOP
      LCTRPY=ISUM
      ISUM=ISUM+NLAY
      LCBOT=ISUM
      ISUM=ISUM+NRC*NBOT
      LCHY=ISUM
      ISUM=ISUM+NRC*NBOT
      LCTOP=ISUM
      ISUM=ISUM+NRC*NTOP
C
C6-----PRINT THE AMOUNT OF SPACE USED BY THE BCF PACKAGE.
      ISP=ISUM-ISOLD
      WRITE(IOUT,101) ISP
      101 FORMAT(1X,I8,' ELEMENTS IN X ARRAY ARE USED BY BCF')
      ISUM1=ISUM-1
      WRITE(IOUT,102) ISUM1,LENX
      102 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
      IF(ISUM1.GT.LENX) WRITE(IOUT,103)
      103 FORMAT(1X,' ***X ARRAY MUST BE DIMENSIONED LARGER***')
C
C7-----RETURN
      RETURN
      END

```


List of Variables for Module BCF1AL

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
I	Module	Index.
IBCFCB	Package	Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will be not be printed or recorded. < 0, flow from each constant-head cell will be printed whenever ICBCFL is set.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISIZ	Module	Number of cells in the grid.
ISOLD	Package	Before this module allocates space, ISOLD is set equal to ISUM. After allocation, ISOLD is subtracted from ISUM to get ISP, the amount of space in the X array allocated by this module.
ISP	Module	Number of words in the X array allocated by this module.
ISS	Package	Flag. = 0, simulation is transient. ≠ 0, simulation is steady state.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1	Module	ISUM-1.
L	Module	Temporary storage for LAYCON(I).
LAYCON	Package	DIMENSION (80) Layer type code: 0 - Layer strictly confined. 1 - Layer strictly unconfined. 2 - Layer confined/unconfined (transmissivity is constant). 3 - Layer confined/unconfined (transmissivity varies).
LCBOT	Package	Location in the X array of the first element of array BOT.
LCHY	Package	Location in the X array of the first element of array HY.
LCSC1	Package	Location in the X array of the first element of array SC1.
LCSC2	Package	Location in the X array of the first element of array SC2.
LCTOP	Package	Location in the X array of the first element of array TOP.
LCTRPY	Package	Location in the X array of the first element of array TRPY.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program.
NBOT	Module	Counter for the number of layers which need elevation of the bottom. Layers for which LAYCON = 1 or 3.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NRC	Module	Number of cells in a layer.
NROW	Global	Number of rows in the grid.
NTOP	Module	Counter for the number of layers which need elevation of the top. LAYCON = 2 or 3.